# $Id$
# Maintainer: Eric Bélanger <eric@archlinux.org>

pkgname=beep
pkgver=1.3
pkgrel=4
pkgdesc="Advanced PC speaker beeping program"
arch=('x86_64')
url="http://www.johnath.com/beep"
license=('GPL2')
depends=('glibc')
source=(http://www.johnath.com/beep/${pkgname}-${pkgver}.tar.gz)
sha512sums=('3f71146202208fa0c0cda360d125dd972ea003cf8cd62da21d67eb544c92f7b82640e7ce7c3835cbc65c75909a882edfa54dbd8d8b748d6f243711ea331477fd')

prepare() {
  cd ${pkgname}-${pkgver}
  sed -i "s:-Wall:${CFLAGS}:" Makefile
}

build() {
  cd ${pkgname}-${pkgver}
  make
}

package() {
  cd ${pkgname}-${pkgver}
  install -D -m755 beep "${pkgdir}/usr/bin/beep"
  install -D -m644 beep.1.gz "${pkgdir}/usr/share/man/man1/beep.1.gz"
}
